

NAG Library Routine Document

F08UBF (DSBGVX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08UBF (DSBGVX) computes selected eigenvalues and, optionally, eigenvectors of a real generalized symmetric-definite banded eigenproblem, of the form

$$Az = \lambda Bz,$$

where A and B are symmetric and banded, and B is also positive definite. Eigenvalues and eigenvectors can be selected by specifying either all eigenvalues, a range of values or a range of indices for the desired eigenvalues.

2 Specification

```
SUBROUTINE F08UBF (JOBZ, RANGE, UPLO, N, KA, KB, AB, LDAB, BB, LDBB, Q, &
                  LDQ, VL, VU, IL, IU, ABSTOL, M, W, Z, LDZ, WORK, &
                  IWORK, JFAIL, INFO)
INTEGER           N, KA, KB, LDAB, LDBB, LDQ, IL, IU, M, LDZ, &
                  IWORK(5*N), JFAIL(*), INFO
REAL (KIND=nag_wp) AB(LDAB,*), BB(LDBB,*), Q(LDQ,*), VL, VU, ABSTOL, &
                  W(N), Z(LDZ,*), WORK(7*N)
CHARACTER(1)     JOBZ, RANGE, UPLO
```

The routine may be called by its LAPACK name *dsbgvx*.

3 Description

The generalized symmetric-definite band problem

$$Az = \lambda Bz$$

is first reduced to a standard band symmetric problem

$$Cx = \lambda x,$$

where C is a symmetric band matrix, using Wilkinson's modification to Crawford's algorithm (see Crawford (1973) and Wilkinson (1977)). The symmetric eigenvalue problem is then solved for the required eigenvalues and eigenvectors, and the eigenvectors are then backtransformed to the eigenvectors of the original problem.

The eigenvectors are normalized so that

$$z^T A z = \lambda \quad \text{and} \quad z^T B z = 1.$$

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Crawford C R (1973) Reduction of a band-symmetric generalized eigenvalue problem *Comm. ACM* **16** 41–44

Demmel J W and Kahan W (1990) Accurate singular values of bidiagonal matrices *SIAM J. Sci. Statist. Comput.* **11** 873–912

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Wilkinson J H (1977) Some recent advances in numerical linear algebra *The State of the Art in Numerical Analysis* (ed D A H Jacobs) Academic Press

5 Arguments

- 1: JOBZ – CHARACTER(1) *Input*
On entry: indicates whether eigenvectors are computed.
 JOBZ = 'N'
 Only eigenvalues are computed.
 JOBZ = 'V'
 Eigenvalues and eigenvectors are computed.
Constraint: JOBZ = 'N' or 'V'.
- 2: RANGE – CHARACTER(1) *Input*
On entry: if RANGE = 'A', all eigenvalues will be found.
 If RANGE = 'V', all eigenvalues in the half-open interval (VL, VU] will be found.
 If RANGE = 'I', the ILth to IUth eigenvalues will be found.
Constraint: RANGE = 'A', 'V' or 'I'.
- 3: UPLO – CHARACTER(1) *Input*
On entry: if UPLO = 'U', the upper triangles of *A* and *B* are stored.
 If UPLO = 'L', the lower triangles of *A* and *B* are stored.
Constraint: UPLO = 'U' or 'L'.
- 4: N – INTEGER *Input*
On entry: *n*, the order of the matrices *A* and *B*.
Constraint: $N \geq 0$.
- 5: KA – INTEGER *Input*
On entry: if UPLO = 'U', the number of superdiagonals, k_a , of the matrix *A*.
 If UPLO = 'L', the number of subdiagonals, k_a , of the matrix *A*.
Constraint: $KA \geq 0$.
- 6: KB – INTEGER *Input*
On entry: if UPLO = 'U', the number of superdiagonals, k_b , of the matrix *B*.
 If UPLO = 'L', the number of subdiagonals, k_b , of the matrix *B*.
Constraint: $KA \geq KB \geq 0$.
- 7: AB(LDAB,*) – REAL (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array AB must be at least $\max(1, N)$.
On entry: the upper or lower triangle of the *n* by *n* symmetric band matrix *A*.
 The matrix is stored in rows 1 to $k_a + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in $AB(k_a + 1 + i - j, j)$ for $\max(1, j - k_a) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in $AB(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_a)$.

On exit: the contents of AB are overwritten.

8: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F08UBF (DSBGVX) is called.

Constraint: LDAB \geq KA + 1.

9: BB(LDBB,*) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array BB must be at least $\max(1, N)$.

On entry: the upper or lower triangle of the n by n symmetric positive definite band matrix B .

The matrix is stored in rows 1 to $k_b + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of B within the band must be stored with element B_{ij} in $BB(k_b + 1 + i - j, j)$ for $\max(1, j - k_b) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of B within the band must be stored with element B_{ij} in $BB(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_b)$.

On exit: the factor S from the split Cholesky factorization $B = S^T S$, as returned by F08UFF (DPBSTF).

10: LDBB – INTEGER *Input*

On entry: the first dimension of the array BB as declared in the (sub)program from which F08UBF (DSBGVX) is called.

Constraint: LDBB \geq KB + 1.

11: Q(LDQ,*) – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array Q must be at least $\max(1, N)$ if JOBZ = 'V', and at least 1 otherwise.

On exit: if JOBZ = 'V', the n by n matrix, Q used in the reduction of the standard form, i.e., $Cx = \lambda x$, from symmetric banded to tridiagonal form.

If JOBZ = 'N', Q is not referenced.

12: LDQ – INTEGER *Input*

On entry: the first dimension of the array Q as declared in the (sub)program from which F08UBF (DSBGVX) is called.

Constraints:

if JOBZ = 'V', LDQ \geq $\max(1, N)$;
otherwise LDQ \geq 1.

13: VL – REAL (KIND=nag_wp) *Input*

14: VU – REAL (KIND=nag_wp) *Input*

On entry: if RANGE = 'V', the lower and upper bounds of the interval to be searched for eigenvalues.

If RANGE = 'A' or 'I', VL and VU are not referenced.

Constraint: if RANGE = 'V', VL < VU.

- 15: IL – INTEGER *Input*
 16: IU – INTEGER *Input*

On entry: if RANGE = 'T', the indices (in ascending order) of the smallest and largest eigenvalues to be returned.

If RANGE = 'A' or 'V', IL and IU are not referenced.

Constraints:

if RANGE = 'T' and $N = 0$, $IL = 1$ and $IU = 0$;
 if RANGE = 'T' and $N > 0$, $1 \leq IL \leq IU \leq N$.

- 17: ABSTOL – REAL (KIND=nag_wp) *Input*

On entry: the absolute error tolerance for the eigenvalues. An approximate eigenvalue is accepted as converged when it is determined to lie in an interval $[a, b]$ of width less than or equal to

$$ABSTOL + \epsilon \max(|a|, |b|),$$

where ϵ is the *machine precision*. If ABSTOL is less than or equal to zero, then $\epsilon \|T\|_1$ will be used in its place, where T is the tridiagonal matrix obtained by reducing C to tridiagonal form. Eigenvalues will be computed most accurately when ABSTOL is set to twice the underflow threshold $2 \times X02AMF()$, not zero. If this routine returns with INFO = 1 to N, indicating that some eigenvectors did not converge, try setting ABSTOL to $2 \times X02AMF()$. See Demmel and Kahan (1990).

- 18: M – INTEGER *Output*

On exit: the total number of eigenvalues found. $0 \leq M \leq N$.

If RANGE = 'A', $M = N$.

If RANGE = 'T', $M = IU - IL + 1$.

- 19: W(N) – REAL (KIND=nag_wp) array *Output*

On exit: the eigenvalues in ascending order.

- 20: Z(LDZ, *) – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array Z must be at least $\max(1, N)$ if JOBZ = 'V', and at least 1 otherwise.

On exit: if JOBZ = 'V', Z contains the matrix Z of eigenvectors, with the i th column of Z holding the eigenvector associated with $W(i)$. The eigenvectors are normalized so that $Z^T B Z = I$.

If JOBZ = 'N', Z is not referenced.

- 21: LDZ – INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08UBF (DSBGVX) is called.

Constraints:

if JOBZ = 'V', $LDZ \geq \max(1, N)$;
 otherwise $LDZ \geq 1$.

- 22: WORK(7 × N) – REAL (KIND=nag_wp) array *Workspace*

- 23: IWORK(5 × N) – INTEGER array *Workspace*

24: JFAIL(*) – INTEGER array Output

Note: the dimension of the array JFAIL must be at least $\max(1, N)$.

On exit: if JOBZ = 'V', then

if INFO = 0, the first M elements of JFAIL are zero;

if INFO = 1 to N, JFAIL contains the indices of the eigenvectors that failed to converge.

If JOBZ = 'N', JFAIL is not referenced.

25: INFO – INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO = 1 to N

If INFO = i , then i eigenvectors failed to converge. Their indices are stored in array JFAIL. Please see ABSTOL.

INFO > N

F08UFF (DPBSTF) returned an error code; i.e., if INFO = $N + i$, for $1 \leq i \leq N$, then the leading minor of order i of B is not positive definite. The factorization of B could not be completed and no eigenvalues or eigenvectors were computed.

7 Accuracy

If B is ill-conditioned with respect to inversion, then the error bounds for the computed eigenvalues and vectors may be large, although when the diagonal elements of B differ widely in magnitude the eigenvalues and eigenvectors may be less sensitive than the condition of B would suggest. See Section 4.10 of Anderson *et al.* (1999) for details of the error bounds.

8 Parallelism and Performance

F08UBF (DSBGVX) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08UBF (DSBGVX) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is proportional to n^3 if JOBZ = 'V' and RANGE = 'A', and assuming that $n \gg k_a$, is approximately proportional to $n^2 k_a$ if JOBZ = 'N'. Otherwise the number of floating-point operations depends upon the number of eigenvectors computed.

The complex analogue of this routine is F08UPF (ZHBGVX).

10 Example

This example finds the eigenvalues in the half-open interval $(0.0, 1.0]$, and corresponding eigenvectors, of the generalized band symmetric eigenproblem $Az = \lambda Bz$, where

$$A = \begin{pmatrix} 0.24 & 0.39 & 0.42 & 0 \\ 0.39 & -0.11 & 0.79 & 0.63 \\ 0.42 & 0.79 & -0.25 & 0.48 \\ 0 & 0.63 & 0.48 & -0.03 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 2.07 & 0.95 & 0 & 0 \\ 0.95 & 1.69 & -0.29 & 0 \\ 0 & -0.29 & 0.65 & -0.33 \\ 0 & 0 & -0.33 & 1.17 \end{pmatrix}.$$

10.1 Program Text

```

Program f08ubfe

!      F08UBF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: dsbgvx, nag_wp, x04caf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Real (Kind=nag_wp), Parameter      :: zero = 0.0E+0_nag_wp
Integer, Parameter                 :: nin = 5, nout = 6
Character (1), Parameter           :: uplo = 'U'
!      .. Local Scalars ..
Real (Kind=nag_wp)                 :: abstol, vl, vu
Integer                             :: i, ifail, il, info, iu, j, ka, kb,      &
                                     ldab, ldbb, ldq, ldz, m, n
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable    :: ab(:,,:), bb(:,,:), q(:,,:), w(:,)    &
                                     work(:,), z(:,)
Integer, Allocatable                :: iwork(:), jfail(:)
!      .. Intrinsic Procedures ..
Intrinsic                           :: max, min
!      .. Executable Statements ..
Write (nout,*) 'F08UBF Example Program Results'
Write (nout,*)
!      Skip heading in data file
Read (nin,*)
Read (nin,*) n, ka, kb
ldab = ka + 1
ldbb = kb + 1
ldq = n
ldz = n
m = n
Allocate (ab(ldab,n),bb(ldbb,n),q(ldq,n),w(n),work(7*n),z(ldz,m),      &
          iwork(5*n),jfail(n))

!      Read the lower and upper bounds of the interval to be searched,
!      and read the upper or lower triangular parts of the matrices A
!      and B from data file

Read (nin,*) vl, vu
If (uplo=='U') Then
  Read (nin,*)((ab(ka+1+i-j,j),j=i,min(n,i+ka)),i=1,n)
  Read (nin,*)((bb(kb+1+i-j,j),j=i,min(n,i+kb)),i=1,n)
Else If (uplo=='L') Then
  Read (nin,*)((ab(1+i-j,j),j=max(1,i-ka),i),i=1,n)
  Read (nin,*)((bb(1+i-j,j),j=max(1,i-kb),i),i=1,n)
End If

!      Set the absolute error tolerance for eigenvalues. With abstol
!      set to zero, the default value is used instead

abstol = zero

!      Solve the generalized symmetric eigenvalue problem

```

```

!      A*x = lambda*B*x

!      The NAG name equivalent of dsbgvx is f08ubf
      Call dsbgvx('Vectors','Values in range',uplo,n,ka,kb,ab,ldab,bb,lddb,q, &
        ldq,vl,vu,il,iu,abstol,m,w,z,ldz,work,iwork,jfail,info)

      If (info>=0 .And. info<=n) Then

!      Print solution

      Write (nout,99999) 'Number of eigenvalues found =', m
      Write (nout,*)
      Write (nout,*) 'Eigenvalues'
      Write (nout,99998) w(1:m)
      Flush (nout)

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04caf('General',' ',n,m,z,ldz,'Selected eigenvectors',ifail)

      If (info>0) Then
        Write (nout,99999) 'INFO eigenvectors failed to converge, INFO =', &
          info
        Write (nout,*) 'Indices of eigenvectors that did not converge'
        Write (nout,99997) jfail(1:m)
      End If
      Else If (info>n .And. info<=2*n) Then
        i = info - n
        Write (nout,99996) 'The leading minor of order ', i, &
          ' of B is not positive definite'
        Write (nout,99999) 'Failure in DSBGVX. INFO =', info
      End If

99999 Format (1X,A,I5)
99998 Format (3X,(8F8.4))
99997 Format (3X,(8I8))
99996 Format (1X,A,I4,A)
      End Program f08ubfe

```

10.2 Program Data

F08UBF Example Program Data

```

4      2      1      :Values of N, KA and KB

0.0    1.0      :Values of VL and VU

0.24   0.39   0.42
      -0.11   0.79   0.63
           -0.25   0.48
           -0.03 :End of matrix A

2.07   0.95
      1.69  -0.29
           0.65  -0.33
           1.17 :End of matrix B

```

10.3 Program Results

F08UBF Example Program Results

```

Number of eigenvalues found =    1

Eigenvalues
  0.0992
Selected eigenvectors

```

	1
1	0.6729
2	-0.1009
3	0.0155
4	-0.3806
